

# **L<sup>A</sup>T<sub>E</sub>X and the Gnuplot Plotting Program**

**David Kotz**

Principal author of this tutorial for gnuplot 3.0, July 3, 1991

**current gnuplot team**

Updates of this tutorial for gnuplot 4.0, March 2004

Consequently, this document describes gnuplot version 3.0 and later.

## **Contents**

<b>1</b>	<b>Introduction and History</b>	<b>1</b>
<b>2</b>	<b>Using gnuplot for L<sup>A</sup>T<sub>E</sub>X: a Tutorial</b>	<b>1</b>
2.1	Summary — Use with L <sup>A</sup> T <sub>E</sub> X . . . . .	6
<b>3</b>	<b>Use with EEPIC</b>	<b>8</b>
<b>4</b>	<b>For Former GnuT<sub>E</sub>X Users</b>	<b>8</b>
<b>5</b>	<b>What is new for T<sub>E</sub>X and L<sup>A</sup>T<sub>E</sub>X terminals in gnuplot 4.0</b>	<b>8</b>
<b>6</b>	<b>Contact for help</b>	<b>8</b>

## 1 Introduction and History

Gnuplot was originally developed by Colin Kelley and Thomas Williams in 1986 to plot functions and data files on a variety of terminals. In 1988 and 1989 I created an alternate version, known as Gnu $\TeX$ , that supported a new “terminal type” called `latex`, so gnuplot would output  $\LaTeX$  code. The plot could then be included in a  $\LaTeX$  document. I added a number of embellishments, supported only by the `latex` terminal, allowing the user to produce publication-quality plots.

In late 1989 and early 1990 Gnu $\TeX$  and a number of other gnuplot variants were merged together into a new release of gnuplot, 2.0. This includes, among many other improvements, a  $\LaTeX$  driver derived from the one in Gnu $\TeX$ . Former Gnu $\TeX$  users are referred to Section 4 for information about adapting to gnuplot. Anyone interested in using gnuplot with  $\LaTeX$  should read the next section, a tutorial, and the primary gnuplot manual.

The reader should note that the  $\LaTeX$  picture environments output by gnuplot can be quite large and complicated, and can easily exceed the memory capacity of  $\TeX$ . If an enlarged version of  $\TeX$  is available, it is wise to use it. Otherwise, keep your plots simple and add `\clearpage` to your document where necessary.

There is also a new EEPIC driver (`eePIC`), intended for use with the EEPIC macro package for  $\LaTeX$ . EEPIC allows for much more efficient line-drawing, runs through  $\LaTeX$  faster, and uses less memory. See Section 3 for more information.

There is a small package of auxiliary files (makefiles and scripts) that I find useful for making  $\LaTeX$  plots with gnuplot. This is available for ftp as `pub/gnuplot-latex.shar` from `cs.duke.edu`. I can mail copies (see the end of this paper for information).

## 2 Using gnuplot for $\LaTeX$ : a Tutorial

Gnuplot is by nature an interactive program. Users making plots for  $\LaTeX$  will generally not use gnuplot interactively. Whenever hard copy is desired from gnuplot, the program need not be run on a graphics terminal. In this case the output is directed to a file or pipe, then sent to the appropriate output device. For example, output from the terminal type `unixplot` may be sent to a program interpreting the Unix plotting standard. The terminal types `imagen` and `postscript` may be used for output to printers understanding those languages. (A shell script (`lasergnu`) is supplied with the distribution that will accept a gnuplot command or input file and send the output to an Imagen or Postscript laser printer. This script may have been adapted to your site.) The terminal type `fig` outputs FIG code that can be read by the Fig graphics program and translated into forms usable in both  $\TeX$  and  $\LaTeX$  documents.


We now ignore the interactive nature of gnuplot and provide the input to gnuplot from a file, *i.e.*,

```
gnuplot plotcommands.gp
```

In this example, all of the commands to gnuplot are contained in the file `plotcommands.gp`. Multiple filenames may be supplied to gnuplot this way, read in the order they are given. The output (one or more plots) may be piped to another program or redirected to a file. Usually, however, we direct the output explicitly with an instruction to gnuplot (the `set output "outfile.tex"` command). Gnuplot continues to print error messages to the terminal (`stderr`). After printing, the output file has to be closed by `set output`, *i.e.* without the file name specification.

**Example 1:** Here is a first example, producing a plot for this document. The gnuplot input file is given below, and the output appears as Figure 1. The input file defines the output to be in  $\LaTeX$ , gives a file name for the output, and plots  $y = \sin(x)$  for  $x$  on  $[-\pi, \pi]$ . To produce the figure, I simply `\input{eg1}` in a `center` environment in a `figure` environment. In following examples, I will enclose the figure in a box to make it look a little better.

```
set terminal latex
set output "eg1.tex"
plot [-3.14:3.14] sin(x)
```

Figure 1: A first example:  $y = \sin(x)$ .

Note that gnuplot has drawn in the axes, labeled the tic marks for us, scaled the  $y$  axis automatically, and added a key in the upper-right-hand corner (this may be moved with the `set key` command, and removed with `unset key`<sup>1</sup>).

This is the default line style for the  $\LaTeX$  driver. Because of the limited picture capabilities of  $\LaTeX$ , many dots are required to approximate drawing a solid line. This may overload the memory of many  $\TeX$  implementations. There are other line types available that draw dotted lines and use much less memory. The EEPIC driver draws solid lines with much less memory usage.

**Example 2:** Now we will embellish the plot a little with some labels. This input file produces Figure 2.


```
set terminal latex
set output "eg2.tex"
set size 5/5., 4/3.
set format xy "%g%"
set title "This is a plot of  $y = \sin(x)$ "
set xlabel "This is the  $x$  axis"
set ylabel "This is the  $y$  axis"
plot [0:6.28] [0:1] sin(x)
```

We have specified the plot to be 5 inches wide and 4 inches tall with the `set size` command. This is the size of the area used by the plot, *including* space for the labels. In the first example, this size was the default 5 inches by 3 inches. By specifying the scaling factors of 1 (or 5/5) and 1.3333 (or 4/3), we obtain the desired plot size.

We have requested that the format used by the  $x$ - and  $y$ -axis tic mark labels be in  $\LaTeX$  math mode. This makes the labels look a little better. The default is `set format xy "%g"`. The `%g` represents the general-purpose floating point formatting specification for the `printf` function in C. Any valid floating-point formatting specification, or  $\LaTeX$  command, is allowed in the format.

A title for the plot and labels for the axes were set up in the next three commands. Note that they are processed by  $\LaTeX$  and so may have math mode and other symbols in them. The `ylabel` may have multiple lines, delineated with `\\`. The `ylabel` can be moved around with optional offset parameters (see `set ylabel` in the gnuplot manual). Typically, the `ylabel` needs to be moved to the left to avoid interfering with the left-hand side of the plot. Once these labels are set up, they will be used for all subsequent plot commands until they are changed. These labels are also supported by the other terminal types, but (of course) any  $\LaTeX$  code in the string will not be interpreted. We have also defined the range of both  $x$  (now  $[0, 2\pi]$ ) and  $y$  (here  $[0, 1]$ ).

<sup>1</sup>In gnuplot version 4.0, the syntax `set noXXX` changed to `unset XXX`.



So far we have plotted one curve,  $y = \sin(x)$ , on one plot. In gnuplot, each `plot` command generates a new plot. If the output is to a screen, the screen is cleared. If to a printer, a new page is produced. In the `latex` case, a new picture is started. It is not likely that  $\LaTeX$  users will want this to happen, so generally each plot has its own input file and is kept in a separate output (`.tex`) file for inclusion at different places in the document.

**Example 3:** To place more than one curve on a plot, use one `plot` statement and separate the description of each curve by a comma. In our next example, we will plot both a function and a data file on the same plot. This plot is shown in Figure 3.

```
set terminal latex
set output "eg3.tex"
set format xy "%g%"
set title "This is another plot"
set xlabel "$x$ axis"
set ylabel "$y$ axis"
set key 15,-10
plot x with lines, "eg3.dat" with linespoints
```




Figure 3: An example with two curves on the same plot.

Here you will see that the  $x$  range was not specified. The  $x$  range is determined automatically, unless specified by the user. In this case, it is defined by the range of the data file "eg3.dat". The function is plotted over the same range. If no data files or  $x$  range are supplied, the default range of  $[-10 : 10]$  is used. We have also moved the key to a different position. The function  $y = x$  is plotted "with lines", which is the default plot style for functions, and is shown here to illustrate the plot style option. The data file `eg3.dat` is plotted with style `linespoints`, a style like `lines` that also plots a symbol at each data point.

There is a style called `points` that only plots the symbols at data points, and another called `dots` that plots a tiny dot for each data point. The `points` and `linespoints` styles produce a different point symbol for each curve on the plot (for up to twelve symbols, after which they are re-used; see Figure 7 for a complete list). The `lines` and `linespoints` styles use a different line style for each curve on the plot (in this example the dots have different spacing). The style `impulses` draws a perpendicular from each point to the  $x$ -axis. Finally, the `errorbars` style can draw error bars at each data point (see the gnuplot manual).

**Example 4:** In the above plots of  $\sin(x)$ , it would make more sense to label the axis in units of  $\pi$ . The position and labels of the tic labels may be specified by the user, with the `set xtics` and `set ytics` commands. This is demonstrated by the following example, shown in Figure 4.

```

set terminal latex
set output "eg4.tex"
set format y "%g"
set format x "%.2f"
set title "This is  $\sin(x)$ "
set xlabel "This is the  $x$  axis"
set ylabel " $\sin(x)$ "
unset key
set xtics -pi, pi/4
plot [-pi:pi] [-1:1] sin(x)

```




Figure 4: An example of the `set xtics` command.

Since `pi` is a predefined variable in gnuplot, we can use it anywhere we may use an expression. The `set xtics` command here specifies that the tics on the  $x$  axis start at  $-\pi$  and increment by  $\pi/4$ . Since no end point is given, the tics continue to the right edge. We have also turned off the key, and changed the format to restrict the  $x$ -axis tic labels to 2 decimal places.

With a little more work, the plot can look even better. Another form of this command allows us to specify the label and position of each tic individually. Replacing the above `set xtics` command with the following gives us Figure 5. We also make use of the line continuation character, the backslash (`\`), to spread out this command for readability.

```

set xtics (" $-\pi$ " -pi,\
" $-\frac{\pi}{2}$ " -pi/2,\
"0" 0,\
" $\frac{\pi}{2}$ " pi/2,\
" $\pi$ " pi)


```

**Going further:** You should now be able to make a variety of plots for your L<sup>A</sup>T<sub>E</sub>X document. We will present a final example without explanation that showcases some of the capabilities of gnuplot. You may find documentation for the various commands in the gnuplot manual, though hopefully this example is somewhat self-explanatory. This is shown in Figure 6.

```

set terminal latex
set output "eg6.tex"
set size 3.5/5, 3/3.
set format y "%g"
set format x "%5.1f $\mu$ "
set title "This is a title"

```



```

set xlabel "This is the  $x$  axis"
set ylabel "This is\\a longer\\version\\ of\\the  $y$ \\ axis"
set label "Data" at -5,-5 right
set arrow from -5,-5 to -3.3,-6.7
set key -4,8
set xtic -10,5,10
plot [-10:10] [-10:10] "eg3.dat" title "Data File" with linespoints 1 7, \
    3*exp(-x*x)+1 title " $3e^{-x^2}+1$ " with lines 4

```

**Line and point types:** For reference, we show all of the line and point types available in Figure 7.

## 2.1 Summary — Use with $\LaTeX$

In summary, to use the  $\LaTeX$  facilities of gnuplot, the first command to gnuplot should be

```
set terminal latex
```

and the output of gnuplot should be directed to a file, for example,

```
set output "plot.tex"
```

This may be anything you like but it should have a `.tex` extension, of course. Then the size of the plot should be given. For example, the command



```
set size 1,2
```

tells gnuplot to use a 5 inch wide by 6 inch high box for the plot. The numbers given are *scale factors*, not the actual size. The default is 5 inches by 3 inches. This is the size of the complete plot, including all labels.

Then you do the `(s)plot`, and finally issue commands to close the file and switch the terminal back to the default by

```
set output
set terminal pop
```

Finally, the file will contain all of the plots you have specified (you probably only want one plot per file). This file can then be used in a  $\LaTeX$  document, *e.g.*,



```

\begin {figure}
  \begin{center}
    \input{plot}
  \end{center}
\end {figure}

```

This puts the plot into a figure.

You will also want to read about the following commands: `set title`, `set xlabel`, `set ylabel`, `set key`, `set label`, `set xtics`, `set ytics`, and `set clip`. These are all described in the regular gnuplot manual.

### 3 Use with EEPIC

EEPIC is a macro package extending the picture environment of L<sup>A</sup>T<sub>E</sub>X. If you have the EPIC or EEPIC macros, and your dvi translator supports the *tpic* \specials, then you can save L<sup>A</sup>T<sub>E</sub>X memory. With EEPIC pictures, the `plot.tex` file will be smaller, L<sup>A</sup>T<sub>E</sub>X will run much faster (and need much less memory), and the dvi file will be smaller. The quality of the output is about the same. If you change the source, you can generate some more interesting line styles.

To use EEPIC, set gnuplot's terminal type to `eepic` instead of `latex`, and use gnuplot as before. The line styles will change. Include the file `plot.tex` in your document as before, along with the document style options `[epic,eepic]`.

### 4 For Former GnuT<sub>E</sub>X Users

Former GnuT<sub>E</sub>X users may be pleased with many of the new features (many inspired by your suggestions!), but will also find many changes. gnuplot will *not* run all GnuT<sub>E</sub>X input files unchanged. Several GnuT<sub>E</sub>X features were not included in gnuplot because they were specific to the L<sup>A</sup>T<sub>E</sub>X driver. I encourage you to use the newer gnuplot. A translator is available that attempts to translate your old GnuT<sub>E</sub>X 1.6 input files into gnuplot 3.0 files. You can ftp it from `cs.duke.edu` as `dist/sources/gnuplot/gnut2p.tar.Z`. This file also contains directions and a list of changes from GnuT<sub>E</sub>X to gnuplot.

### 5 What is new for T<sub>E</sub>X and L<sup>A</sup>T<sub>E</sub>X terminals in gnuplot 4.0

In addition to the `latex` terminal, the following L<sup>A</sup>T<sub>E</sub>X-friendly terminals are available:

- `emtex`: Like the `latex` terminal, but supports emtex specials: any line slopes contrary to a very limited set of L<sup>A</sup>T<sub>E</sub>X slopes.
- `epslatex`: Combined L<sup>A</sup>T<sub>E</sub>X and postscript parts for text and lines, respectively, with the postscript part included by `\includegraphics{...}` command.
- `pstex` and `pslatex`: Combined T<sub>E</sub>X / L<sup>A</sup>T<sub>E</sub>X and postscript parts for text and lines, respectively, included by `\special{psfile=...}` command.
- `mf` and `mp`: Produces metafont and metapost outputs.

See helps of these terminals for more details about their usage.

In addition, the `postscript eps enhanced` is the most useful for T<sub>E</sub>X and L<sup>A</sup>T<sub>E</sub>X if you don't insist of using T<sub>E</sub>X fonts for the graph labels, and `pdf enhanced` and `png enhanced` for pdf<sup>L</sup>A<sup>T</sup>E<sup>X</sup>.

### 6 Contact for help

For general gnuplot questions, the gnuplot mailing list `info-gnuplot@lists.sourceforge.net` is where you can send your e-mail, or you can use gnuplot newsgroup `comp.graphics.apps.gnuplot`. Additional sources of information are available on gnuplot homepage `www.gnuplot.info`.